A Simple Model

August 3, 2015

1 The Model

A representative household's problem is

$$max_{\{c_t,l_t\}_{t=0}^{\infty}} E_0 \sum_{t=1}^{\infty} \beta^{t-1} \frac{(c_t^{\theta}(1-l_t)^{1-\theta})^{1-\tau}}{1-\tau} \tag{1}$$

subject to the resource constraint

$$c_t + i_t = e^{z_t} k_t^{\alpha} l_t^{1-\alpha} = y_t \tag{2}$$

the law of motion for capital

$$k_{t+1} = i_t + (1-\delta)k_t \tag{3}$$

and the stochastic process for productivity

$$z_t = \rho z_{t-1} + s\epsilon_t \tag{4}$$

with $\epsilon_t \sim N(0, \sigma^2)$.

The FOC of the model w.r.t consumption and labor supply as follows:

$$\frac{\left(c_t^{\theta}\left(1-l_t\right)^{1-\theta}\right)^{1-\tau}}{c_t} = \beta E_t \left(\frac{\left(c_{t+1}^{\theta}\left(1-l_{t+1}\right)^{1-\theta}\right)^{1-\tau}}{c_{t+1}} \left(1+\alpha e^{z_t} k_t^{\alpha} l_t^{1-\alpha} - \delta\right) \right) \tag{5}$$

$$\frac{1-\theta}{\theta} \frac{c_t}{1-l_t} = (1-\alpha) e^{z_t} k_t^{\alpha} l_t^{-\alpha} \tag{6}$$

An equilibrium is characterized by a system of 6 equations: (2-6) above with equation (2) has two equal signs.

2 The Code

You can input the FOC equations (variables in level or in log-level) and also log-linearized equations into Dynare mod file. All are acceptable. Here we use FOC equations both in level and log-level. If log-linearized equations input, then you should change the keyword 'model;' into 'model(linear);' to tell Dynare you have input log linearization equations.[1].

We typically want to approximate the solutions of natural logs of the variables, so that impulse responses, etc. are all in percentage terms. Dynare will do linear approximation of the levels of the variables. To get it to do linear approximation of the logs of the variables, you need to specify the variables as "exp(x)". This way the variable x is interpreted as the log of the variable of interest while exp(x) is the level (since the exponential and log are inverse functions) which is what actually shows up in the most of the FOCs. Then you just type the FOCs.

In Dynare, the timing of the variable reflects the date when the variable is decided. For instance, the capital stock for time t is decided in time $t-1$(end of period). In order to exploit the syntax of Dynare, the timing for capital stock in capital accumulation equation is modified[2].

2.1 Level

See the timing of the capital stock in the model.

```
% This file is modified by Li Xiangyang @ 2012-5-6 to run in Dynare 4.2.0 or above
var c k lab z;
varexo e;

parameters bet the del alp tau rho s;

bet     = 0.987;
the     = 0.357;
del     = 0.012;
alp     = 0.4;
tau     = 2;
```

[1]The following is my understanding, but I am not so sure if it is correct: if the model is declared to be linear, then initval block no longer needed, the endogenous variables are automatically assigned zeroes as their steady state values since it is the essence of the log-linearization.

[2]Please do not name the mod file using only numbers or the same with variables name in memory/workplace in Matlab.

```
rho     = 0.95;
s       = 0.007;

model;
% define model-local variables to improve readability;
# mar_c= (c^the*(1-lab)^(1-the))^(1-tau);
# mar_c1=(c(+1)^the*(1-lab(+1))^(1-the))^(1-tau);

%(1) Euler equation
mar_c/c=bet*(mar_c1/c(+1))*(1+alp*exp(z)*k(-1)^(alp-1)*lab^(1-alp)-del);

%(2) wage equation
c=the/(1-the)*(1-alp)*exp(z)*k(-1)^alp*lab^(-alp)*(1-lab);

%(3)capital accumulation equation
k=exp(z)*k(-1)^alp*lab^(1-alp)-c+(1-del)*k(-1);

%(4) technology shock
z=rho*z(-1)+e;
end;

initval;
k   = 1;
c   = 1;
lab = 0.3;
z   = 0;
e   = 0;
end;

shocks;
var e=s^2;
end;

steady;

%if periods not specify, there will be no simulations.
stoch_simul(periods=1000,irf=40,simul_seed=123456,order=1);

%save the simulated data to file
dynasave('simudata.mat');
```

2.2 Log-level

```
% This file is modified by Li Xiangyang @ 2013-8-30 to run in Dynare 4.2.0 or above
var y c k i lab z;
varexo e;
```

```
parameters bet the del alp tau rho s;

bet     = 0.987;
the     = 0.357;
del     = 0.012;
alp     = 0.4;
tau     = 2;
rho     = 0.95;
s       = 0.007;

model;
% define model-local variables to improve readability;
# mar_c = (exp(c)^the*(1-exp(lab))^(1-the))^(1-tau);
# mar_c1= (exp(c(+1))^the*(1-exp(lab(+1)))^(1-the))^(1-tau);

%(1) Euler equation
mar_c/exp(c)=bet*(mar_c1/exp(c(+1)))*(1+alp*exp(z)*exp(k(-1))^(alp-1)*exp(lab)^(1-alp)-del);

%(2) wage equation
exp(c)=the/(1-the)*(1-alp)*exp(z)*exp(k(-1))^alp*exp(lab)^(-alp)*(1-exp(lab));

%(3) capital accumulation equation
exp(k)=exp(i)+(1-del)*exp(k(-1));

%(4) the production technology
exp(y)=exp(z)*exp(k(-1))^alp*exp(lab)^(1-alp);

%(5) the resource constraint
exp(y)=exp(c)+exp(i);

%(6)the technology shock
z=rho*z(-1)+e;
end;

initval;
k   = log(2.71828);
c   = log(2.71828);
lab = log(exp(0.3));
z   = log(1);
e   = log(1);
end;

steady;
check;
```

```
shocks;
var e=s^2;
end;



%if periods not specify, there will be no simulations.
stoch_simul(periods=1000,irf=40,simul_seed=123456,order=1);

%save the simulated data to file
dynasave('simudata_exp.mat');
```